


Graphs and Relations


1 A Graph of Things

Remember the universe of discs and triangles from the first tutorial? We still have the universe



```
data Thing = R|S|T|U|V|W|X|Y|Z
things = [ R, S, T, U, V, W, X, Y, Z ]
```

and predicates `isRed`, `isGreen`, `isBlue`, `isSmall`, `isBig`, `isDisc`, `isTriangle`. We add a relation to this universe. We say “*X* likes *Y*” if there is an arrow from *X* to *Y*. Here is the universe with the relation: We call this a directed graph.



This relation is defined for you in [Tutorial5.hs](https://gavinphr.github.io/INF1A/tutorial5/Tutorial5.hs)¹, implemented as the function:

```
likes :: Thing -> Thing -> Bool -- x `likes` y iff there is an arrow x -> y
```

```
GHCi> likes R T
True
GHCi> likes R U
False
```

It is often clearer to use ``likes`` in infix style, with backquotes (```):

```
GHCi> R `likes` T
True
GHCi> R `likes` U
False
```

¹<https://gavinphr.github.io/INF1A/tutorial5/Tutorial5.hs>

For each of the following statements, determine whether it is true by just looking at the graph and then use Haskell to verify your answer.

1. Some thing likes S .

Solution: True.

```
GHCi> or [x `likes` S | x <- things]
True
```

2. Every thing likes W .

Solution: False.

```
GHCi> and [x `likes` W | x <- things]
False
```

3. Every thing is liked by some thing.

Solution: True.

```
GHCi> and [or [y `likes` x | y <- things] | x <- things]
True
```

4. Some thing is not liked by any thing.

Solution: False.

```
GHCi> or [not (or [y `likes` x | y <- things]) | x <- things]
False
```

What is the relationship between statement 3 and 4?

2 Quantifiers and Formal Representation

Natural languages such as English and Chinese are intrinsically ambiguous. Statements in natural languages can be interpreted in many different ways. To avoid ambiguity (and for other benefits), let's try to formally represent the statements in the previous exercise.

First we need the two **quantifiers**:

1. forall \forall
2. exists \exists

So for the first statement, “**Something** likes S .”, we write:

$$\exists x \in \text{things} (x \text{ `likes` } S)$$

It reads: there exists an x in **things** such that x `likes` S .

For the third statement, “**Everything** is liked by **something**.”, we write:

$$\forall x \in \text{things} (\exists y \in \text{things} (y \text{ `loves` } x))$$

Convince yourself that the formal statement conveys the same meaning as the original.

A benefit of formal statements is that we can easily turn them into Haskell code. We first need two auxiliary functions to represent \forall and \exists :

```
every :: [t] -> (t -> Bool) -> Bool
every xs p = and [p x | x <- xs]
```

```
some :: [t] -> (t -> Bool) -> Bool
some ys p = or [p y | y <- ys]
```

You will often find it useful to use a λ -function or a section as the predicate p .

For example, $\exists y \in \text{things } (y \text{ `likes` } S)$ may be translated to:

```
some things (\y -> y `likes` S)
GHCi> some things (\y -> y `likes` S)
True
GHCi> some things (`likes` S) -- we can also use a section
True
```

And $\forall x \in \text{things } (\exists y \in \text{things } (y \text{ `loves` } x))$ is translated to:

```
every things (\x -> some things (\y -> y `likes` x))
every things (\x -> some things (`likes` x)) -- again we can use a section
GHCi> every things (\x -> some things (`likes` x))
True
```

Translate the rest of the statements, first to formal mathematical representations, then to Haskell.

5. Every thing likes W .

Solution: $\forall x \in \text{things } (x \text{ `likes` } W)$

```
every things (\x -> x `likes` W)
```

6. Some thing is not liked by any thing.

Solution: $\exists y \in \text{things } \neg(\exists x \in \text{things } (x \text{ `likes` } y))$

```
some things (\y -> not(some things (`likes` y)))
 $\exists y \in \text{things } (\forall x \in \text{things } \neg(x \text{ `likes` } y))$ 
some things (\y -> every things (\x -> not (x `likes` y)))
some things (\y -> every things (neg (`likes` y)))
```

3 A Social Graph

In the first exercise, we could answer all of the questions by just looking at the graph. In this exercise, the graph is much larger. We have 100 people, given as a list of type `[Person]` :

```
data Person = Carmina|Malika|Margherita|Ginny|Loida|Rosalie|Nikki
             -- the list continues ...

people :: [Person]
people = [Carmina, Malika, Margherita, Ginny, Loida, Rosalie, Nikki,
         Bea, Delores, Soledad, Sheba, Elsy, Lashanda,Carolynn, .....
```

Each person can be:

- either happy or not happy.
- either a scientist or not a scientist.
- either an artist or not an artist.
- either a politician or not a politician.

You can check whether a person has a particular property using the predicates:

```
isHappy      :: Person -> Bool      isArtist      :: Person -> Bool
isScientist  :: Person -> Bool      isPolitician  :: Person -> Bool
```

For example, we can find out whether Carmina is happy:

```
GHCi> isHappy Carmina
False
```

`:-` (. Observe that one can have more than 1 property i.e. a person can be happy, a scientist, an artist, and a politician at the same time.

There is a *loves* relation in our social graph that is similar to the *likes* relation in the first exercise. We can check whether Carmina loves Brian using the function `loves :: Person -> Person -> Bool`:

```
GHCi> Carmina `loves` Brian
True
```

For each of the statements below, first turn it into a formal mathematical statement, then translate the formal statement into Haskell code and determine whether the statement is true. A couple of examples are already done for you.

7. Everybody loves somebody.

Solution: $\forall x \in \text{people} \exists y \in \text{people} (x \text{ `loves` } y)$

```
every people (\x -> some people (\y -> x `loves` y))
```

8. Somebody loves everybody. (In English, this statement is ambiguous. Give two formalisations, one for each possible meaning and check whether either is true in our universe.

Solution: $\exists x \in \text{people} \forall y \in \text{people} (x \text{ `loves` } y)$

```
 $\forall y \in \text{people} \exists x \in \text{people} (x \text{ `loves` } y)$ 
```

```
some people (\x -> every people (\y -> x `loves` y))
every people (\y -> some people (\x -> x `loves` y))
```

9. Everybody loves someone who loves them.

Solution: $\forall x \in \text{people} \exists y \in \text{people} (x \text{ loves } y \wedge y \text{ loves } x)$
`every people (\x -> some people (\y -> (x `loves` y) && (y `loves` x)))`

10. Somebody loves someone who loves them.

Solution: $\exists x \in \text{people} \exists y \in \text{people} (x \text{ loves } y \wedge y \text{ loves } x)$
`some people (\x -> some people (\y -> (x `loves` y) && (y `loves` x)))`

11. Everybody (x), loves some y , who loves some z , who loves them (x).²

Solution:
 $\forall x \in \text{people} \exists y \in \text{people} \exists z \in \text{people} (x \text{ loves } y \wedge y \text{ loves } z \wedge z \text{ loves } x)$
`every people (\x -> some people (\y -> some people
(\z -> (x `loves` y) && (y `loves` z) && (z `loves` x))))`

12. Every politician loves an artist.

Solution: $\forall x \in \text{Politicians} (\exists y \in \text{Artists} (x \text{ loves } y))$
`every [x | x <- people, isPolitician x] (\x -> some
[y | y <- people, isArtist y] (\y -> x `loves` y))`

13. Everyone loved by an artist loves a scientist.

Solution: $\forall x \in \text{People loved by an artist} (\exists s \in \text{Scientists} (x \text{ loves } s))$
`every [x | x <- people, some [a | a <- people, isArtist a]
(\a -> a `loves` x)] (\x -> some [s | s <- people, isScientist s]
(\s -> x `loves` s))`

14. Some artist has only one love.

Note that a common trick for expressing the “only one” constraint is by decomposing it into two constraints: “at least one” and “no more than one”.

Solution: $\exists a \in \text{Artists} (\exists x \in \text{people} ((a \text{ loves } x) \wedge \forall y \neq x (\text{not } (a \text{ loves } y))))$
`some [a | a <- people, isArtist a] (\a -> some people
(\x -> a `loves` x && every [y | y <- people, y /= x]
(\y -> not (a `loves` y))))`

²In other words, everyone belongs to some love triangle.

*Six degrees of separation is the idea that all people are six, or fewer, social connections away from each other. It is claimed that, a chain of "a friend of a friend" statements can be made to connect any two people in a maximum of six steps.*³ In this exercise, you will find the degree of separation of our social graph, also called the *diameter* of the graph.

In our case, the social connection is the `loves` relation. Let's first think about what it means for a person `x` to be able to reach another person `z` in one step, it is simply that:

```
reach1 :: Person -> Person -> Bool
reach1 x z = x `loves` z
```

How can we make a function that tells us whether person `x` is able to reach person `z` in 2 steps? If there is a person `y` such that `x` loves (i.e. `x` can reach `y` in 1 step) and `y` can reach `z` in 1 step, then `x` can reach `z` in 2 steps.

How can we tell whether `x` can reach `z` in 3 steps? If there is a `y` such that `x` likes and `y` can reach `z` in 2 steps, then `x` can reach `z` in 3 steps.

15. Now implement the functions `reach2` and `reach3`.

Solution:

```
reach2 :: Person -> Person -> Bool
reach2 x z = some people (\y -> x `loves` y && reach1 y z)

reach3 :: Person -> Person -> Bool
reach3 x z = some people (\y -> x `loves` y && reach2 y z)
```

16. What is the degree of separation in our graph?

Solution: 3

³https://en.wikipedia.org/wiki/Six_degrees_of_separation

4 Sudoku

A site listing [all known 17-clue sudoku problems](#) with a unique solution encodes a sudoku as a string of 81 digits. Each character represents a square on the board. The first nine characters represent the first row, and so on; 0 represents a blank square, and a digit 1..9 represents an entry in the corresponding square. The file Tutorial5.hs includes three examples of proposed solutions, and a function

```
entries :: String -> Int -> Int -> Int -> Bool
entries str i j k
  | length str /= 81 || or [ not (isDigit c) | c <- str ] = error "Ill-formed string"
  | or [ arg < 1 || arg > 9 | arg <- [i,j,k] ] = error "index out of range [1..9]"
  | otherwise = str!!(9 * (i - 1) + (j - 1)) == intToDigit k
```

that converts such a string prob into a predicate `entered = entries prob`, taking three integer arguments in the range `[0..9]` – `entered i j k` is `True` iff the digit `k` is *entered* in square (i, j) .

Your task is to complete the following function to check whether the examples really are solutions. Do they satisfy the rules of sudoku?

```
check :: (Int -> Int -> Int -> Bool) -> Bool
check entered =
  -- every square is filled
  and [ or[entered i j k | k <- [1..9] ]
        | i <- [1..9], j <- [1..9] ]
  && -- no square is filled twice
  and [ or[ not(entered i j k), not (entered i j k') ]
        | i <- [1..9], j <- [1..9], k <- [1..9], k' <- [1..(k-1)] ]
  && -- every row contains every digit
  and [ or [entered i j k | j <- [1..9] ]
        | i <- [1..9], k <- [1..9] ]
  -- every column contains every digit
  -- every big square contains every digit
```

17. Complete the code to check for the two remaining properties, and check the three examples.

```
check :: (Int -> Int -> Int -> Bool) -> Bool
check entered =
  -- every square is filled
  and [ or[entered i j k | k <- [1..9] ]
        | i <- [1..9], j <- [1..9] ]
  && -- no square is filled twice
  and [ or[ not(entered i j k), not (entered i j k') ]
        | i <- [1..9], j <- [1..9], k <- [1..9], k' <- [1..(k-1)] ]
  && -- every row contains every digit
  and [ or [entered i j k | j <- [1..9] ]
        | i <- [1..9], k <- [1..9] ]
  && -- every column contains every digit
  and [ or [entered i j k | i <- [1..9] ]
        | j <- [1..9], k <- [1..9] ]
  && -- every big square contains every digit
  and [ or [entered (3*p+q) (3*r+s) k | q <- [1..3], s <- [1..3] ]
        | p <- [0..2], r <- [0..2], k <- [1..9] ]
```